

Paper 107-2010

Challenge! Reading Mainframe Hex Delimited Flat File Where Each Line Has Different Layout

Anjan Matlapudi and Knapp, J. Daniel
Pharmacy Informatics, PerformRx, The Next Generation PBM,
Philadelphia, PA

ABSTRACT

This paper illustrates how to design an appropriate input program to handle a complex file layout using data collected from pharmacy and health insurance information about individuals. Various INFILE and INPUT options are illustrated in the process, and some related functions are considered.

The input file is the output of a COBOL program pulling data from a DB2 database which is then brought to the PC via FTP. Each record contains 6 types of information, called segments, for a person. The segments and the fields within are divided by unprintable hexadecimal codes, which SAS represents with notations like the hexadecimal numbers 1E (Segment separator) and 1C (Field separator) respectively. A further complication is the use of 1D which is the group separator to separate repeating segments. There are also groups of repeating fields within a segment. Since the fields do not have a fixed length and they may be missing on some records, there is no fixed record layout for the file.

Although the program was written for the PC, the technique is applicable for any system. All the tools discussed are in BASE SAS®. The typical attendee or reader will have some experience in SAS, but not a lot of experience dealing with the input of external data.

INTRODUCTION

The SAS® System has excellent facilities for importing and manipulating data between platforms. This presentation is designed to review how to convert a hierarchical EBCDIC (Extended Binary Coded Decimal Interchange Code) text file from an IBM mainframe source into a set of SAS datasets on an ASCII (American Standard Code for Information Interchange) system. The examples in this paper explain where and how to use the INFILE and INPUT statements. Alternative approaches using character handling functions are also discussed.

THE FILE STRUCTURE

The file begins with a header record and ends with a trailer that we will exclude. The remaining records are divided into segments of related fields. Each record starts with '02' and ends with '03'. Below we have shown two data records.¹

The first segment, known as G1, is exceptional and will be eliminated from our discussion. The other segments begin with 1E (Dec 30), but may be preceded by 1D (Dec 29) indicating that that segment may repeat within one record (see segment 'AM07'). Segments are further divided into fields that begin with 1C (Dec 28). Each field is divided into a two character identifier followed by the value of the field. For example, 'AM' is the identifier and '04' is the value for segment 'AM04'. The next field in 'AM04' has the identifier, C2 (in dark red) and value, '3424342355', which is the identifier of the person whose information is on this record. CC gives the first name and CD the last.

```

Header Segment
RULE:  ↓-----1-----2-----3-----4-----5-----6-----7-----8-----
1  CHAR  .00T4454831521009339070001  0648558201008020648P11622760 . 58
    ZONE  033533333333333333333333223333333333333333333333333333320
    NUMR  2004445483152100933907000100064855820100802064801162276003

    Start of Text
2  CHAR  .G1608191393160076051B1445483152  1011487752192      20090727PAN00104 . AM04.C23424342355 Line 1
    ZONE  0433333333333333333333333333333333333333333222233333333354433333221144331433333333333
    NUMR  27160819139316007605121445483152010114877521920000002009072701E0010400EC01D0C323424342355
  
```

¹ The real data files contained 5 to 10 million data records.

The FILENAME statement ties the FILEREf, INP, to the input physical file location. It separates this information from the DATA step; hence it is easier to modify. In addition it provides various options to provide the flexibility to handle many different data sources. The second FILENAME statement does the same for the output file.

The following DATA step creates the new file with segment structure. It illustrates handling external file I/O. No SAS datasets are output in this step. One should not create SAS datasets when none are needed because this activity takes up a significant portion of the execution time for a DATA step.

```
*----break up record into segments discarding unused segments----*;
data _null_ ;
  length seg $200 segnm $4 ;
  infile inp dlm='1E'x truncover lrecl=400 firstobs=2 ;
  do i = 1 to 6 ;
    input seg :$char200. @ ;
    segnm = substr (seg , 2, 4) ;
  *----segments are skipped to simplify the code for this paper.----*;
    if segnm in: ("G1" "99" "AM07" "AM11") then ;
  else
    do ;
      seg = translate(seg, " " , "1D"x ) ;
      file temp ;
      put seg ;
    end ;
  end ;
run;
```

The INFILE statement associates the FILEREf, INP, with an input buffer. In addition, it provides options for flexibility. The DLM option, which specifies the "field" separator², is most important to our task, since the segments are separated by 1E³. Otherwise one must use character handling functions to manipulate the data.

TRUNCOVER is the contemporary way to handle variable length files. It accepts that data that is there without going to the next record looking for a field that is short. The default value is FLOWOVER.⁴

When records are longer than the default length, 256 bytes, then the LRECL option is required. It determines the longest length that the buffer should need to hold a complete record.

FIRSTOBS was set to 2 in order to skip over the header record.

The FILE statement does for output what the INFILE statement does for input.

In addition to providing a place to specify options, the INFILE and FILE statements are executable. They set the current buffer. In the above program, INPUT applies to FILEREf INP and its associated buffer instead of the default FILEREf DATALINES. Similarly PUT applies to TEMP instead of the default LOG. The concept is important because it means that you can read from multiple input buffers and write to multiple output buffers.

The INPUT statement does the actual reading, in this case, from the INP file because of the preceding INFILE statement. There are three types of input, LIST INPUT, COLUMN INPUT, and FORMATTED INPUT. Usually formatted input is the most appropriate choice for complex reading, but in this case we have a modified form of list input because the separator 1E determines the length of the segment. The use of a format is a direct command to read the specified number of byte, i.e. ignoring separators when required. The colon in front of \$CHAR200. says to respect the separator, i.e. stop reading when the separator is encountered. It provides a handy way to have the correct combination of formatted input with list input.

The trailing @-signs says to hold the input record until either the next INPUT statement applied to that buffer or the bottom of the implied loop of the DATA step (whichever comes first). Consequently the same record is read in the DO-loop 6 times. The limit is 6 because there are always exactly 6 segments in the files that this program reads.

² For the purposes of this step, the whole segment is a field.

³ SAS uses an "X" to refer to hexadecimal numbers. For example, 1CX indicates the decimal 28. In character form this is "1C"X.

⁴ FLOWOVER goes to the next record when a field is short. MISOVER throws away short the short field and does not go to the next record. It is important for historical reasons, but usually TRUNOVER is the option of choice.

There is also a double trailing @ which holds a record over the iteration of the DATA step. In all but the simplest of situations it is probably best to avoid the double trailing @ because it leaves dirty messages in the log.

The special segments G1 and '99' are skipped in this program. G1 occurs at the beginning of every data record as a sort of header to a record, and '99' is the trailer at the end of the file.

The TRANSLATE function is used to eliminate the code 1D. No segment was repeated within an input record, so the code was just a nuisance and this was a good time to get rid of it since this simplifies the next step.

In this step both the reading and writing are done in an explicit loop. The common idiom in SAS is one record in one out with no explicit loop. Why is this different? We want to change the structure of the data. Although one could force the SAS idiom, it would make the code more error prone and harder to read. One should learn to write DATA steps that differ from the natural rhythm of SAS when the data structure calls for it.

STEP2 – OUTPUT A SAS DATA SET FOR EACH SEGMENT

The DATA step is organized by subroutines – one for each segment. This helps to make the step easier to read and modify since the code associated with a segment is in one place. The function of the main routine is to control the whole process.

The main routine consists of the DATA statement with the list of output datasets, a LENGTH statement including all the variables, and a SELECT block to decide which subroutine to call for current segment. If an unknown code is encountered, the program aborts.

The subroutines consist of INPUT and OUTPUT statements for that segment plus any special code needed to handle that segment. For example, segment AM05 can have repeating fields. The INPUT statement assumes one repeat and a CHECK field to prove there are no more. When CHECK is not missing, the step aborts. The TRUNCOVER option allows including the repeat variables in the AM05 routine, since they will be made blank if they aren't there and will be filled in if they are. This subroutine structure makes the program easy to read and modify.

The two preceding paragraphs illustrate one technique for checking the input assumptions to preserve the integrity of the program and the programmer.

The hex number 1C is the field separator and the first two bytes are the field identifier.

We use +2 to skip over this portion of the field and the colon modified INFORMAT to preserve the LIST INPUT nature of the file as explained in the previous DATA step.

```
*----write out wanted segments ----*;
data am04 (keep = seq_num CardH_Id fname lname Ecl_code Group_num)
    am07 (keep = seq_num Rx_Service_RefNum_Quali Rx_Service_refNum
                Product_Service_ID_Qualifier Product_Service
                Quantity_Despedsed Fill_Number Days_Supply
                Compound_Code DAW_Code Date_Prescription_Written
                Submission_Clarif_Code Submission_Clarif_Code
                Other_Coverage_Code Unit_of_Measure)
    am03 (keep = seq_num Prescriber_ID_Qualifier Prescriber_ID)
    am05 (keep = seq_num COBCount CoverageType IDQualifier PayerID
                PayerDate CoverageType PaidQual AmountPaid1 RejectCount
                RejectCode PaidCount PayerDate2 PaidCount2
                PaidQual2 AmountPaid2 RejectCount2 RejectCode2)
    am11 (keep = seq_num Ingredient_CostDispensing_Fee Patient_Paid_Amount
                Sales_Tax_Amount Usual_Customary_Charge
                Basis_Cost_Determination)

;

    Length segnm                                $ 4
           seq_num                              8
           /* am04 */
           CardH_Id                             $ 20
           fname                               $ 12
           lname                               $ 15
```

```

        Ecl_code                8
        Group_num              $ 8
    /* am03 */
        Prescriber_ID_Qualifier $ 2
        Prescriber_ID          $ 15
    /* am05 */
        COBCount               $ 1
        CoverageType           $ 2
        IDQualifier             $ 2
        PayerID                $ 10
        PayerDate              $ 8
        PaidCount              $ 8
        PaidQual               $ 8
        Amount_Paid1           $ 8
        RejectCount            $ 8
        RejectCode             $ 2
    /* other repeated fields */
        CoverageType2          $ 2
        PayerDate2             $ 8
        PaidCount2             $ 8
        PaidQual2              $ 8
        Amount_Paid2           $ 8
        RejectCount2           $ 8
        RejectCode2            $ 2
    ;
    infile temp dlm = "021C"x truncover ;
    input segnm :$char4. @ ;
    select (segnm) ;
        when ("AM04")do; link am04 ;end ;
        when ("AM03")do; link am03 ;end ;
        when ("AM05")do; link am05 ;end ;
        when ("AM04", "AM03", "AM05") /* skip for now */
            otherwise error "Bad segment name" ;
    end ;
    return ;

*----routine for each segment----*;

AM04:
input
    +2 CardH_Id                :$char20.
    +2 fname                   :$char12.
    +2 lname                   :$char15.
    +2 Ecl_code                 :1.
    +2 Group_num               :$char8.
;
output am04 ;
seq_num + 1 ;
return ;

AM03:
Input
    +2 Prescriber_ID_Qualifier :$char2.
    +2 Prescriber_ID           :$char15.
;

Output am03 ;
seq_num + 1 ;
return ;

AM05:
input
    +2 COBCount                :$Char1.

```

```

+2 CoverageType           :$char2.
+2 IDQualifier            :$char20.
+2 PayerID                :$char10.
+2 PayerDate              :$char8.
+2 PaidCount              :1.
+2 PaidQual               :1.
+2 Amount_Paid1           :$char8.
+2 RejectCount            :1.
+2 RejectCode             :$char2.
/* possible repeat */
+2 CoverageType2          :$char2.
+2 PayerDate2             :$char8.
+2 PaidCount2             :1.
+2 PaidQual2              :1.
+2 Amount_Paid2           :$char8.
+2 RejectCount2           :1.
+2 RejectCode2            :$char2.
/*Convert Over punch sings to dollar ammount*/;
AmountPaid1 = put(input(Amount_Paid1, ZD8.2), 8.2);
AmountPaid2 = put(input(Amount_Paid2, ZD8.2), 8.2)
;
if missing(COBCCount) then abort ;
output am05 ;
seq_num + 1 ;
return ;
;

*** routines for AM07 and AM11 are similar but not shown here ***;
run ;

```

CONCLUSION

We have illustrated how to make the reading of fairly complex external data into a simple program by separating it into two well structured steps. The key here is to recognize the structure of the separators requires introducing two steps instead of trying to do the reading in one step. Steps not mentioned carry out analysis of the data and move it into other systems for use by doctors and pharmacists.

REFERENCES

Zirbel, Doug. Functioning JCL Into a SAS® Relational Database Table: Your Portable Tutorial On Character Functions, Plus an MVS™ Batch Bonus, [Coders' Corner](#) Paper 96-25.

Kuligowski, T. Andrew. Datalines, Sequential Files, CVS, HTML and More – Using INFILE and INPUT Statements to Introduce External Data into the SAS System, [SUGI 31](#) *Tutorials* Paper 228-31.

ACKNOWLEDGMENTS

We like to acknowledge Ian Whitlock for helping us in preparing this paper. We also like to thank Marian Whitlock for sharing her ideas to put all things together.

We did like to acknowledge Mr. Shimels Afework, The Sr. Director of our company. The PerformRx was formed in 1999 as a division of the [AmeriHealth Mercy Family of Companies](#). We provide specialized pharmacy benefit management services (PBM) through proactively managing escalating pharmacy costs while focusing on clinical and financial results.

CONTACT INFORMATION:

Your comments and questions are valued and encouraged. Contact the author at

Name Anjan Matlapudi
 Senior Pharmacy Analyst
Address PerformRx, The Next Generation PBM
 200 Stevens Drive
 Philadelphia, PA 19113
Work Phone: (215)937-7252
Fax: (215)863-5100
E-mail: anjan.matlapudi@performrx.com
 matanjan@hotmail.com

Name Knapp, J. Daniel
 Senior Manager
Address PerformRx, The Next Generation PBM
 200 Stevens Drive
 Philadelphia, PA 19113
Work Phone: (215-937-7251
Fax: (215)863-5100
E-mail: Daniel.Knapp@performrx.com

SAS® is a registered trademark or trademark of SAS® Institute, Inc. in the USA and other countries. IBM, OS/390, and MVS are registered trademarks of International Business Machines Inc.